
1	Introduction
MCData needs support of MSRP (RFC 4975).
 
a) New type definitions are needed.
b) As MSRP uses TCP, a generic TCP handler needs to be added to the IP PTC and InternetApplication_Type needs to be extended.
2	Test Model
2.1	IPCAN Model


[bookmark: _Hlk90465842]2.2	EUTRA Model


2.3	Impact
· TCP handler has no impact on system interface
· For support of MSRP new value needed for InternetApplication_Type 
( No impact as long as support of MSRP is not required)
· For support of MSRP:
· Codec implementation for types shown in clause 3.2
· Support of InternetApplication_Type 'msrp' by IP socket implementation

3	TTCN Changes and new Type Definitions
3.1	InternetApplication_Type
	  type enumerated InternetApplication_Type { ims, http, msrp }; 



3.2	MSRP_TypeDefs
	module MSRP_TypeDefs {

  import from CommonDefs all;
  import from LibSip_Common all;
  import from LibSip_MessageBodyTypes all;
  
  type record MSRP_URI_Authority {              /* RFC3986 clause 3.2: authority = [ userinfo "@" ] host [ ":" port ] */ 
    charstring userInfo optional,
    charstring hostName,
    integer portNumber optional
  };
  
  type record MSRP_URI {                        /* RFC 4975 clause 9: MSRP-URI */
    charstring scheme,                          /* msrp-scheme: "msrp" | "msrps" */
    MSRP_URI_Authority authority,               /* authority as defined in RFC 3986 clause 3.2 */
    charstring sessionId optional,              /* session-id = 1*( unreserved / "+" / "=" / "/" ) */
    charstring transport,                       /* transport = "tcp" / 1*ALPHANUM; mandatory parameter */
    SemicolonParam_List uriParameters           /* URI-parameter = token ["=" token]; empty list when there are no parameters */
  };

  type record of MSRP_URI MSRP_URI_List;
  
  type record MSRP_FromToPath {                 /* RFC 4975 clause 9: To-Path = "To-Path:" SP MSRP-URI *( SP MSRP-URI )
                                                   RFC 4975 clause 9: From-Path = "From-Path:" SP MSRP-URI *( SP MSRP-URI ) */
    MSRP_URI_List msrpUriList
  };

  type union MSRP_IntegerOrWildcard {           /* RFC 4975 clause 9: range-end, total */
    integer val,                                /* 1*DIGIT */
    Null_Type wildcard                          /* "*" */
  };
                                                   
  type record MSRP_ByteRange {                  /* RFC 4975 clause 9: Byte-Range = "Byte-Range:" SP range-start "-" range-end "/" total */
    integer rangeStart,                         /* 1*DIGIT */
    MSRP_IntegerOrWildcard rangeEnd,            /* 1*DIGIT / "*" */
    MSRP_IntegerOrWildcard total                /* 1*DIGIT / "*" */
  };

  type record MSRP_Status {                     /* RFC 4975 clause 9: Status = "Status:" SP namespace SP status-code [SP comment] */
    charstring namespace,                       /* namespace = 3(DIGIT); "000" for all codes defined in RFC 4975 */
    Char3_Type statusCode,                      /* status-code = 3DIGIT */
    charstring comment optional                 /* comment = utf8text */
  };

  type record MSRP_ExtHeader {                  /* RFC 4975 clause 9: ext-header = hname ":" SP hval CRLF */
    charstring hname,
    charstring hval
  };

  type union MSRP_Header {                      /* RFC 4975 clause 9: header = Message-ID / Success-Report / Failure-Report / Byte-Range / Status / ext-header */
    charstring messageId,                       /* Message-ID = "Message-ID:" SP ident */
    charstring successReport,                   /* Success-Report = "Success-Report:" SP ("yes" / "no" ) */
    charstring failureReport,                   /* Failure-Report = "Failure-Report:" SP ("yes" / "no" / "partial" ) */
    MSRP_ByteRange byteRange,                   /* Byte-Range = "Byte-Range:" SP range-start "-" range-end "/" total */
    MSRP_Status status,                         /* Status = "Status:" SP namespace SP status-code [SP comment] */
    MSRP_ExtHeader extHeader                    /* ext-header: may carry any other header not defined in RFC 4975 */
  };

  type record MSRP_Headers {                    /* RFC 4975 clause 9: headers = To-Path CRLF From-Path CRLF 1*( header CRLF ) */
    MSRP_FromToPath toPath,                     /* To-Path: mandatory, has to be first header */
    MSRP_FromToPath fromPath,                   /* From-Path: mandatory, has to be second header */
    set of MSRP_Header headers                  /* 1*( header CRLF ) */
  };

  type record MIME_ExtensionField {             /* RFC 2045 clause 9: Any RFC 822 header field which begins with the string "Content-", or
                                                   RFC 2045 clause 8: "Content-Description" ":" *text */
    charstring fieldName,                       /* RFC 822 clause 3.2: 1*<any CHAR, excluding CTLs, SPACE, and ":"> */
    charstring fieldBody                        /* RFC 822 clause 3.2: field-body  =  field-body-contents [CRLF LWSP-char field-body]; field-body-contents = <the ASCII characters ...> */
  };

  type union MSRP_OtherMimeHeader {             /* Other-Mime-header = (Content-ID / Content-Description / Content-Disposition / mime-extension-field) */
    MIME_ContentId contentId,                   /* RFC 2045 clause 7 */
    MIME_ContentDisposition contentDisposition, /* RFC 2183 */
    MIME_ExtensionField mimeExtensionField      /* MIME-extension-field according to RFC 2045 clause 9 or 
                                                   Content-Description according to RFC 2045 clause 8 */
  };

  type record MSRP_Content {                    /* content-stuff = *(Other-Mime-header CRLF) Content-Type 2CRLF data CRLF */
    set of MSRP_OtherMimeHeader otherMimeHeaders,
    MIME_ContentType contentType,               /* equivalent definition of Content-Type as in RFC 2045 clause 5 */
    octetstring data                            /* data = *OCTET */
  };

  type record MSRP_EndLine {                    /* end-line = "-------" transact-id continuation-flag CRLF */
    charstring transactId,
    Char1_Type continuationFlag                 /* continuation-flag = "+" / "$" / "#" */
  };

  //----------------------------------------------------------------------------

  type record MSRP_RequestStart {               /* RFC 4975 clause 9: req-start = "MSRP" SP transact-id SP method CRLF */
    charstring transactId,                      /* transact-id */
    charstring method                           /* method: SEND | REPORT  */
  };

  type record MSRP_Request {                    /* RFC 4975 clause 9: msrp-request = req-start headers [content-stuff] end-line */
    MSRP_RequestStart requestStart,             /* req-start */
    MSRP_Headers headers,                       /* headers */
    MSRP_Content content optional,              /* [content-stuff] */
    MSRP_EndLine endLine                        /* end-line */
  };

  //----------------------------------------------------------------------------

  type record MSRP_ResponseStart {              /* RFC 4975 clause 9: resp-start = "MSRP" SP transact-id SP status-code [SP comment] CRLF */
    charstring transactId,                      /* transact-id */
    Char3_Type statusCode,                      /* status-code: 3DIGIT */
    charstring comment                          /* comment */
  };

  type record MSRP_Response {                   /* RFC 4975 clause 9: msrp-response = resp-start headers end-line */
    MSRP_ResponseStart responseStart,           /* resp-start */
    MSRP_Headers headers,                       /* headers */
    MSRP_EndLine endLine                        /* end-line */
  };
}

